

Yu Xia's Exam 3

Yu Xia
ID: yx5262

Fall 2022

1 RBC in dynare

a.

Consumers maximize his/her utility:

$$\begin{aligned} & \max \sum_{t=0}^{\infty} \beta^{t+s} U_{t+s} \\ & \text{subject to} \\ & c_t + k_t + T_t = w_t n_t + r_t k_{t-1} + k_{t-1} (1 - \delta). \end{aligned}$$

\Longleftrightarrow

$$\begin{aligned} & \max \sum_{t=0}^{\infty} \beta^t (\log(c_t) - \phi n_t) \\ & \text{subject to} \\ & c_t + k_t + T_t - w_t n_t - r_t k_{t-1} - k_{t-1} (1 - \delta) = 0. \end{aligned}$$

The Lagrangian:

$$\begin{aligned} \mathcal{L} &= \sum_{s=0}^{\infty} \beta^{t+s} ((\log(c_{t+s}) - \phi n_{t+s}) \\ & \quad - \lambda_t (c_{t+s} + k_{t+s} + T_{t+s} - w_{t+s} n_{t+s} - r_{t+s} k_{t+s-1} - k_{t+s-1} (1 - \delta))) \\ &= \dots + (\log(c_t) - \phi n_t) - \lambda_t (c_t + k_t + T_t - w_t n_t - r_t k_{t-1} - k_{t-1} (1 - \delta)) \\ & \quad + \beta (\log(c_{t+1}) - \phi n_{t+1}) \\ & \quad - \beta \lambda_{t+1} (c_{t+1} + k_{t+1} + T_{t+1} - w_{t+1} n_{t+1} - r_{t+1} k_t - k_t (1 - \delta)) + \dots \end{aligned}$$

FOCs:	
w.r.t c_t :	
$\frac{1}{c_t} - \lambda_t = 0$	(1.2)
w.r.t n_t :	
$-\phi + \lambda_t w_t = 0$	(1.2)
w.r.t k_t :	
$-\lambda_t + \beta \lambda_{t+1} (r_{t+1} + (1 - \delta)) = 0$	(1.3)
w.r.t λ_t :	
$c_t + k_t + T_t - w_t n_t - r_t k_{t-1} - k_{t-1} (1 - \delta) = 0$	(1.4)

Derivative w.r.t. c_t :

$$\frac{1}{c_t} - \lambda_t = 0$$

\Longleftrightarrow

$$\frac{1}{c_t} = \lambda_t$$

\Longleftrightarrow

$$c_t = \frac{1}{\lambda_t}$$

In other words, the consumer's consumption choice is limited by the budget constraint.

Derivative w.r.t. c_{t+1} :

$$\frac{1}{c_{t+1}} - \lambda_{t+1} = 0$$

Similar to period t , just replaced by $t + 1$.

\Longleftrightarrow

$$\frac{\frac{1}{c_t}}{\frac{1}{c_{t+1}}} = \frac{c_{t+1}}{c_t} = \frac{\frac{1}{\lambda_{t+1}}}{\frac{1}{\lambda_t}} = \frac{\lambda_t}{\lambda_{t+1}}$$

Derivative w.r.t. n_t :

$$-\phi - \lambda_t (-w_t) = 0$$

\Longleftrightarrow

$$-\phi + \lambda_t w_t = 0$$

Consumer's labor choice is determined by the wage.

Derivative w.r.t. n_{t+1} :

$$-\phi + \lambda_{t+1}w_{t+1} = 0$$

Similar to period t , just replaced by $t + 1$.

$$\phi = \lambda_t w_t$$

$$\Longleftrightarrow$$

$$\frac{\phi}{w_t} = \lambda_t = \frac{1}{c_t}$$

$$w_t = \phi c_t \tag{1.5}$$

Using across periods:

$$\frac{\lambda_t}{\lambda_{t+1}} = \frac{\frac{\phi}{w_t}}{\frac{\phi}{w_{t+1}}} = \frac{\frac{1}{w_t}}{\frac{1}{w_{t+1}}} = \frac{w_{t+1}}{w_t}$$

Derivative w.r.t. λ_t :

$$c_t + k_t + T_t - w_t n_t - r_t k_{t-1} - k_{t-1} (1 - \delta) = 0$$

The budget constraint.

Derivative w.r.t. k_t :

$$-\lambda_t - \beta \lambda_{t+1} (-r_{t+1} - (1 - \delta)) = 0$$

$$\Longleftrightarrow$$

$$\beta \lambda_{t+1} (r_{t+1} + (1 - \delta)) = \lambda_t$$

$$\Longleftrightarrow$$

$$\beta (r_{t+1} + (1 - \delta)) = \frac{\lambda_t}{\lambda_{t+1}}$$

It suggests that the choice of capital is affected by the real interest rate in the next period and depreciation rate, also the budget constraints in different periods.

Plug into the cross period consumption FOCs we have:

$$\frac{\frac{1}{c_t}}{\frac{1}{c_{t+1}}} = \frac{c_{t+1}}{c_t} = \beta (r_{t+1} + (1 - \delta))$$

$$\frac{1}{c_t} = \beta (r_{t+1} + (1 - \delta)) \frac{1}{c_{t+1}} \tag{1.6}$$

Plug into the cross period labor FOCs we have:

$$\frac{1}{w_t} = \beta (r_{t+1} + (1 - \delta)) \frac{1}{w_{t+1}} \quad (1.7)$$

The benefit of waiting one period to consume or work is on the return on capital.

b.

The firm's problem in each period:

$$\begin{aligned} & \max_{k_{t-1}, n_t^d} y_t - r_t k_{t-1} - w_t n_t^d \\ \Leftrightarrow & \max_{k_{t-1}, n_t^d} A_t (k_{t-1})^\alpha (n_t^d)^{1-\alpha} - r_t k_{t-1} - w_t n_t^d \end{aligned}$$

FOCs:

w.r.t k_{t-1} :

$$A_t \alpha \left(\frac{k_{t-1}}{n_t^d} \right)^{\alpha-1} = r_t \quad (1.8)$$

w.r.t n_t^d :

$$A_t (1 - \alpha) \left(\frac{k_{t-1}}{n_t^d} \right)^\alpha = w_t \quad (1.9)$$

Derivative w.r.t. k_{t-1} :

$$A_t \alpha (k_{t-1})^{\alpha-1} (n_t^d)^{1-\alpha} - r_t = 0$$

\Leftrightarrow

$$A_t \alpha \left(\frac{k_{t-1}}{n_t^d} \right)^{\alpha-1} = r_t$$

Derivative w.r.t n_t^d :

$$A_t (k_{t-1})^\alpha (1 - \alpha) (n_t^d)^{-\alpha} - w_t = 0$$

\Leftrightarrow

$$A_t (1 - \alpha) \left(\frac{k_{t-1}}{n_t^d} \right)^\alpha = w_t$$

c.

Labor market clearing condition:

$$n_t = n_t^d \quad (1.10)$$

Capital clearing condition:

$$k_t^d = k_{t-1} \quad (1.11)$$

GDP identity:

$$c_t + k_t + g_t = A_t (k_{t-1})^\alpha (n_t^d)^{1-\alpha} + k_{t-1} (1 - \delta) \quad (1.12)$$

In the dynare file:

Endogenous variables: $c_t, k_{t-1}, k_t^d, n_t, n_t^d, r_t, w_t, y_t, A_t, g_t$.

Exogenous variables: $e_{g,t}, e_t$, the government expenditure shock and TFP shock.

Parameter variables: $\beta, \delta, \alpha, \theta, \rho, \phi$.

d.

$$\frac{1}{c_t} = \beta (E_t r_{t+1} + (1 - \delta)) E_t \frac{1}{c_{t+1}} \quad (1.13)$$

$$w_t = \phi c_t \quad (1.5)$$

$$A_t \alpha \left(\frac{k_{t-1}}{n_t} \right)^{\alpha-1} = r_t \quad (1.14)$$

$$A_t (1 - \alpha) \left(\frac{k_{t-1}}{n_t} \right)^\alpha = w_t \quad (1.15)$$

$$c_t + k_t + g_t = A_t (k_{t-1})^\alpha (n_t)^{1-\alpha} + k_{t-1} (1 - \delta) \quad (1.16)$$

$$y_t = A_t (k_{t-1})^\alpha (n_t)^{1-\alpha} \quad (1.17)$$

$$g_t = \theta g_{t-1} + e_{g,t} \quad (1.18)$$

$$\log(A_t) = \rho \log(A_{t-1}) + e_t \quad (1.19)$$

Steady state:

$$A = 1 \quad (1.20)$$

$$g = 0 \quad (1.21)$$

$$r = \frac{1}{\beta} - (1 - \delta) \quad (1.22)$$

$$c = \frac{r}{\phi} \cdot \frac{1 - \alpha}{\alpha} \cdot \left(\frac{r}{\alpha}\right)^{\frac{1}{\alpha-1}} \quad (1.23)$$

$$w = \phi c \quad (1.24)$$

$$n = \frac{c}{\left(\frac{r}{\alpha}\right)^{\frac{\alpha}{\alpha-1}} - \delta \left(\frac{r}{\alpha}\right)^{\frac{1}{\alpha-1}}} \quad (1.25)$$

$$k = \frac{\alpha}{1 - \alpha} \cdot \frac{w}{r} \cdot n \quad (1.26)$$

$$y = n^{1-\alpha} k^\alpha \quad (1.27)$$

e.

```
var c k n r w y a g; // Define endogenous variables
varexo eg ea; // Define exogenous variables
parameters beta delta alpha rhog rhoa phi; // Define parameters variables
alpha = 0.3; // capital share
beta = 0.95; // discount rate
phi = 0.6; // parameter in the consumer's utility: scale parameter
delta = 0.025; // capital depreciation rate
rhog = 0.9; // auto-correlation of g
rhoa = 0.9; // auto-correlation of a

model;
1/c=beta*(1/(c(+1)))*(r(+1)+1-delta);
w = phi*c;
r = alpha*a*(k(-1))^(alpha-1)*n^(1-alpha);
w = (1-alpha)*a*(k(-1))^alpha*n^(-alpha);
k+c+g = (k(-1))*(1-delta)
+a*(k(-1))^alpha*n^(1-alpha);
log(a) = rhoa*log(a(-1))+ea;
y = a*(k(-1))^alpha*n^(1-alpha);
g = rhog*g(-1)+eg;
end;

steady_state_model;
a = 1;
g = 0;
r = 1/beta - (1-delta);
c = r/phi * (1-alpha)/alpha * (r/alpha)^(1/(alpha-1));
w = phi*c;
n = c / ( (r/alpha)^(alpha/(alpha-1))
- delta*(r/alpha)^(1/(alpha-1)));
k = alpha/(1-alpha) * w/r *n;
y = n^(1-alpha)*k^(alpha);
end;

steady; // calculate the steady state

shocks;
var ea; // type of the shock
stderr 0.01; // Scale of the shock
var eg; // type of the shock
stderr 0.01; // Scale of the shock
end;

check;
stoch_simul(order=1,irf=40,hp_filter=1600);
```

f.

```

clc;
clear;
close all;

%-----%
%% Attempt 1: Take Home Q1
dynare Q1_v1.mod;

```




Figure 1: Orthogonalized shock to $e_{g,t}$

As shown above, all the endogenous variables come into steady state gradually.

Consumption and wage decrease at the beginning and then come into a steady state.

Labor, real interest rate, output y_t and of course government expenditure go up at first, and converge to steady state.

Capital goes up gradually at a decreasing speed, then goes to steady state slowly.




Figure 2: Orthogonalized shock to e_t

As shown above, all the endogenous variables come into steady state gradually.

Consumption, capital and wage go up in the beginning then go to a steady state.

Labor and real interest rate increase at the response of the shock, then reduce quickly, even go below the steady state, and converge to steady state eventually.

TFP, output goes up at the beginning, then monotonically converges to the steady state.

In this model, government expenditure shock produces counterfactual predictions for consumption (compared to data). But the TFP shock in this model matches the data: both labor and consumption are procyclical.

2 NKPC in dynare

a.

The first equation describes the trade-off between consumption and labor.

The second equation describes the inter-period trade-off of consumption and saving.

The third equation comes from the firm side, is the NKPC derived from the wholesale firm's and retail firm's optimization problem.

The fourth equation is the wholesale firm's technology we defined.

The fifth equation is the resource constraint. It also implies that the market is clear.

The last equation describes that the monetary authority follows a version of the Taylor rule.

b.

Endogenous variables: $c_t, n_t, i_t, w_t, y_t, \pi_t$ in each period.

We consider the monetary policy shock e .

c.

In steady state:

$$\phi n^\gamma c = w \quad (2.1)$$

$$1 = \beta \frac{1+i}{1+\pi} \quad (2.2)$$

$$\psi \pi (1+\pi) = \frac{1}{1-\epsilon} (1-\epsilon w) n + \beta \psi \pi (1+\pi) \quad (2.3)$$

$$y = n \quad (2.4)$$

$$n = c + \frac{\psi}{2} \pi^2 \quad (2.5)$$

$$\pi = 0 \quad (2.6)$$

Plug (2.6) into previous equations we have:

$$\phi n^\gamma c = w \quad (2.7)$$

$$1 = \beta (1+i) \quad (2.8)$$

$$0 = \frac{1}{1-\epsilon} (1-\epsilon w) n \quad (2.9)$$

$$y = n \quad (2.10)$$

$$n = c \quad (2.11)$$

$$\pi = 0 \quad (2.12)$$

Since $1-\epsilon \neq 0$, $n \neq 0$, (2.9) becomes:

$$1-\epsilon w = 0$$

$$w = \frac{1}{\epsilon}$$

Substitute the system of equations we get:

$$\phi c^\gamma c = w = \frac{1}{\epsilon}$$

$$\Longleftrightarrow$$

$$\phi c^{\gamma+1} = \frac{1}{\epsilon}$$

$$\Longleftrightarrow$$

$$c^{\gamma+1} = \frac{1}{\phi \epsilon}$$

$$c = \frac{1}{(\phi \epsilon)^{\frac{1}{\gamma+1}}} \quad (2.13)$$

$$1+i = \frac{1}{\beta} \quad (2.14)$$

$$w = \frac{1}{\epsilon} \quad (2.15)$$

$$y = n \quad (2.16)$$

$$n = c \quad (2.17)$$

$$\pi = 0 \quad (2.18)$$

d.

```

var c n i w y pi; // Define endogenous variables
varexo e; // Define exogenous variables
parameters phi gamma beta psi epsilon lambda rho; // Define parameters
phi = 0.6;
gamma = 1;
beta = 0.95; // discount rate
psi = 0.5;
epsilon = 1.2;
lambda = 1.5;
rho = 0.9;

model;
w = phi*c*n^(gamma);
1/c = beta*(1/(c(+1)))*((1+i)/(1+pi(+1)));
psi*pi*(1+pi) = (1/(1-epsilon))*(1-epsilon*w)*n
+ beta*psi*(pi(+1))*(1+(pi(+1)));
y = n;
n = c+(psi/2)*pi^2;
i = rho*(i(-1))+(1-rho)*(1/beta-1)+lambda*pi+e;
end;

steady_state_model;
c = 1/((phi*epsilon)^(1/(gamma+1)));
i = 1/beta-1;
w = 1/epsilon;
n = 1/((phi*epsilon)^(1/(gamma+1)));
y = 1/((phi*epsilon)^(1/(gamma+1)));
pi = 0;
end;

steady; // calculate the steady state

shocks;
var e; // type of the shock
stderr 0.01;
end;

```

```

check;
stoch_simul(order=1,irf=40, hp_filter=1600);

Use dynare:

clc;
clear;
close all;
%-----%
%% Attempt 1: Take Home Q2
dynare Q2_v1.mod;

```




Figure 3: Orthogonalized shock to e

e.

After the shock leads to a temporary increment in nominal interest rate:

Consumption, labor, wage, total output and inflation rate decrease initially, and go to steady state eventually.

Marginal cost is the same as the wage in this case.

It's a good explanation. If the nominal interest rate increases more than the inflation, the real interest rate goes down, thus the economy cooling down.

Also, as we can see, the increase in nominal interest rate leads to a temporary recession: reduction in consumption and output, makes workers unwilling to work, and wage decreases as well.

It shows how important the policy shock is.